«بسمه‌تعالی»

«پاسخ تکلیف شمارهٔ ۱ درس بهینه‌سازی ترکیبیاتی ترم اول ۱۴۰۱-۱۴۰۰»

سوال اول: سوال 37 پایان بخش 9.2 از کتاب Winston

The Indiana University Business School has two rooms that each seat 50 students, one room that seats 100 students, and one room that seats 150 students. Classes are held five hours a day. The four types of requests for rooms are listed in Table 37. The business school must decide how many requests of each type should be assigned to each type of room. Penalties for each type of assignment are given in Table 38. An X means that a request must be satisfied by a room of adequate size. Formulate an IP whose solution will tell the business school how to assign classes to rooms in a way that minimizes total penalties.

TABLE 37

Type	Size Room Requested (Seats)	Hours Requested	Number of Requests
1	50	2, 3, 4	3
2	150	1, 2, 3	1
3	100	5	1
4	50	1, 2	2

TABLE 38

Size Requested	Sizes Used to Satisfy Request			Penalty
	50	100	150	
50	0	2	4	100* (Hours requested)
100	X	0	1	100* (Hours requested)
150	X	X	0	100* (Hours requested)

پاسخ: همانطور که در کلاس گفته شد، فرض می‌کنیم از هر نوع درس گروه‌های مختلفی داریم که بازه‌های زمانی ارائه آنها در هر روز داده شده است و باید تصمیم بگیریم که هر درس را به کدام کلاس تخصیص دهیم. برای دروسی که ارائه نمی‌شوند، جریمه بزرگی در نظر می‌گیریم.

اندیس‌های زیر را تعریف می‌کنیم:

$i = 1, \dots, 7$ اندیس دروس با احتساب گروه‌های مختلف

$t = 1, \dots, 5$ اندیس بازه‌های زمانی

$j = 1,2,3$ اندیس انواع اتاق‌ها

پارامترهای زیر را تعریف می‌کنیم:

p_i: جمعیت در درس i

n_j : تعداد اتاق نوع j

cap_j: ظرفیت اتاق نوع j

$a_{i,t}$: پارامتر باینری که اگر درس i در بازه زمانی t ارائه شود، یک و در غیر این صورت صفر است.

$c_{i,j}$: جریمه تخصیص درس i به اتاق نوع j

i	1	2	3	4	5	6	7
p_i	50	50	50	150	100	50	50

j	cap_j	n_j
1	50	2
2	100	1
3	150	1

$a_{i,t}$

	$t=1$	$t=2$	$t=3$	$t=4$	$t=5$
$i=1$	0	1	1	1	0
$i=2$	0	1	1	1	0
$i=3$	0	1	1	1	0
$i=4$	1	1	1	0	0
$i=5$	0	0	0	0	1
$i=6$	1	1	0	0	0
$i=7$	1	1	0	0	0

$c_{i,j}$

	$j=1$	$j=2$	$j=3$
$i=1$	0	2	4
$i=2$	0	2	4
$i=3$	0	2	4
$i=4$	-	-	0
$i=5$	-	0	1
$i=6$	0	2	4
$i=7$	0	2	4

متغیرهای تصمیم را به صورت زیر تعریف می‌کنیم:

δ_i: متغیر باینری که اگر درس i ارائه شود، یک و در غیر این صورت، صفر است.

$\gamma_{i,j}$: متغیر باینری که اگر درس i به اتاق j تخصیص یابد، یک و در غیر این صورت، صفر است.

مسأله به صورت زیر فرمول‌بندی می‌شود:

$$\min z = M\sum_{i=1}^{7}(1-\delta_i) + \sum_{i=1}^{7}\sum_{j=1}^{3}\left(\left(\sum_{t=1}^{5} a_{i,t}\right) \times 100 \times c_{i,j} \times \gamma_{i,j}\right)$$

$s.t.$

$$\sum_{j=1}^{3}\gamma_{i,j} = \delta_i \quad \forall i = 1, \dots, 7$$

$$\sum_{i=1}^{7} a_{i,t}\,\gamma_{i,j} \le n_j \quad \forall j = 1,2,3, \forall t = 1, \dots, 5$$

$$\gamma_{i,j} = 0 \quad \forall i = 1, \dots, 7, \forall j = 1,2,3{:}\, p_i > cap_j$$

```
option MIP=CPLEX;
set
          i/1*7/
          j/1*3/
          t/1*5/;
parameter
          p(i)/1 50, 2 50, 3 50, 4 150, 5 100, 6 50, 7
50/
          cap(j)/1 50, 2 100, 3  150/
          n(j)/1 2,  2 1, 3 1/
          a(i,t)
          c(i,j)/1.2 2, 1.3 4, 2.2 2, 2.3 4, 3.2 2, 3.3
4, 5.3 1, 6.2 2, 6.3 4,
                 7.2 2, 7.3 4/;

loop(i $ (i.val<=3),
          a(i,'2')=1;
          a(i,'3')=1;
          a(i,'4')=1;
);

a('4','1')=1;
a('4','2')=1;
```

```
a('4','3')=1;
a('5','5')=1;
loop(i $ (i.val>=6),
         a(i,'1')=1;
         a(i,'2')=1;
);

binary variable delta(i), gama(i,j);
variable z;

equation obj, const1, const2, const3;
obj..
         z=e=10000*sum(i,(1-delta(i)))+sum((i,j),
sum(t,a(i,t))*100*c(i,j)*gama(i,j));
const1(i)..
         sum(j, gama(i,j))=e=delta(i);
const2(j,t)..
         sum(i, a(i,t)*gama(i,j))=l=n(j);
const3(i,j)$(p(i)>cap(j))..
         gama(i,j)=e=0;

model problem1 /obj, const1, const2, const3/;
solve problem1 using MIP minimizing z;
display delta.l, gama.l, z.l;
display a, c, cap, p, n;
```

سوال دوم: سوال 38 پایان بخش 9.2 از کتاب Winston

A company sells seven types of boxes, ranging in volume from 17 to 33 cubic feet. The demand and size of each box are given in Table 39. The variable cost (in dollars) of producing each box is equal to the box's volume. A fixed cost of $1,000 is incurred to produce any of a particular box. If the company desires, demand for a box may be satisfied by a box of larger size. Formulate and solve (with LINDO, LINGO, or Excel Solver) an IP whose solution will minimize the cost of meeting the demand for boxes.

TABLE 39

	Box						
	1	2	3	4	5	6	7
Size	33	30	26	24	19	18	17
Demand	400	300	500	700	200	400	200

پاسخ:

اندیس‌های زیر را تعریف می‌کنیم:

$i, j = 1, \dots, 7$ اندیس جعبه‌ها

پارامترهای زیر را تعریف می‌کنیم:

s_i: اندازه جعبه i

d_i: تقاضای جعبه i

f: هزینه ثابت

متغیرهای تصمیم را به صورت زیر تعریف می‌کنیم:

δ_i: اگر خط تولید جعبه i راه‌اندازی شود.

$y_{i,j}$: متغیر عددصحیح و نامنفی بیانگر میزان تقاضای جعبه i که توسط جعبه j تامین می‌شود. $s_i \leq s_j$

مسأله به صورت زیر فرمول‌بندی می‌شود:

$$\min z = f\sum_{i=1}^{7}\delta_i + \sum_{i=1}^{7}\sum_{j=1:s_i\le s_j}^{7} s_j y_{i,j}$$

$$\sum_{i=1:s_i\le s_j}^{7} y_{i,j} \le \left(\sum_{i=1:s_i\le s_j}^{7} d_i\right)\delta_j \quad \forall j=1,\dots,7$$

$$\sum_{j=1:s_i\le s_j}^{7} y_{i,j} = d_i \quad \forall i=1,\dots,7$$

$$y_{i,j}\ge 0, int \quad \forall i,j=1,\dots,7: s_i\le s_j$$

$$\delta_i\in\{0,1\} \quad \forall i=1,\dots,7$$

```
set i/1*7/;
alias (i,j)
option optcr=0;
option MIP=CPLEX;
parameter
         d(i)/1 400, 2 300, 3 500, 4 700, 5 200, 6 400,
7 200/
         s(i)/1 33, 2 30, 3 26, 4 24, 5 19, 6 18, 7 17/
         f/1000/

binary variable delta(i);
integer variable y(i,j);
variable z;
y.up(i,j)=10000;

equation obj, const1, const2;
obj..
         z=e=f*sum(i, delta(i))+sum(i,
sum(j$(s(i)<=s(j)), s(j)*y(i,j)));
const1(j)..

sum(i$(s(i)<=s(j)),y(i,j))=l=sum(i$(s(i)<=s(j)),d(i))*d
```

```
elta(j);
const2(i)..
         sum(j$(s(i)<=s(j)),y(i,j))=e=d(i);
model problem2/obj, const1, const2/;
solve problem2 using MIp minimizing z;
display z.l, delta.l, y.l;
```